AZUL

Jorge Bennasar Vázquez, Enrique Alonso Álvarez

With this case study our attempt was to obtain the optimal solution (maximizing the total number of points) of the game Azul in a 3x3 board. We have done this with a Mixed Integer Linear model. The results (obtained with NEOS and GAMS) have been satisfactory, obtaining an optimal solution of 33 points. Additionally, we have done a modification of the problem in which the central position of the board has to be completed in the last turn. In this case we obtained a total punctuation of 30 points. In this report we show the mathematical formulation of the problem, the GAMS code, a review of the results of the original problem and a brief analysis of an alternative problem.

I. PROBLEM STATEMENT

Azul is a board game in which the goal is to obtain the maximum number of points as a square is sequentially filled in. Let us suppose a 3x3 square, in which points will be obtained as follows:

There are 9 boxes that must be filled in with an X. Whenever a box is filled in, the player will obtain as many points as adjacent rows or columns are formed with that X (taking into account the length of the rows and columns).

II. DESCRIPTION AND HYPOTHESES

We have decided to solve the 3x3 case. The model that we propose to do so is a Mixed Integer Linear model. Each position in the square has a different range of possibilities for obtaining an amount of points depending on the number of X that are placed in the adjacent positions. The model that we propose not only takes into account the adjacent positions completed, but also the possible combinations of them that may affect to the outcome of points. To do this, we have not only created variables for points and completed positions, but also for combinations of positions completed. This is crucial, because it allows the model to give the correct amount of points in each turn.

III. MATHEMATICAL FORMULATION OF THE OPTIMIZATION PROBLEM

A. Sets

$i, o \in \{1, \dots, 9\}$: Turn
$j, k, l, m, n \in \{1, \dots, 9\}$: Position
$U1(j,k)$: Connections between j and k
$U2(j,k,l)$: Connections between j, k and l
$U3(j,k,l,m)$: Connections between j, k, l and m
$U4(j,k,l,m,n)$: Connections between j, k, l, m and n

Example (see *D.7.*): the only combination of *j*, *k*, *l*, *m* and *n* relevant to the problem is $U4(5,2,4,6,8)$, used to calculate E_{i52468}, which is later used in the calculation of p_{i5}. This would be the only "connection" between j, k, l, m and n.

B. Parameters

There are no parameters involved in the problem.

C. Variables

1. Free variables

TP: Total points

2. Positive variables

p_{ij}: Points obtained for completing position *j* in turn *i*

3. Binary variables

A_{ij}: Position *j* completed in turn *i*
B_{ijk}: Position *k* completed before *j* is completed in turn *i*
C_{ijkl}: Positions *k* and *l* completed before *j* is completed in turn *i*
D_{ijklm}: Positions *k*, *l* and *m* completed before *j* is completed in turn *i*
E_{ijklmn}: Positions *k*, *l*, *m* and *n* completed before *j* is completed in turn *i*

D. Equations

1. There is only one turn per position:

$$\sum_i A_{ij} = 1 \quad \forall j$$

2. There is only one position per turn:

$$\sum_j A_{ij} = 1 \quad \forall i$$

October 28TH, 2018

Sets of equations 3, 4, 5 and 6 are used to calculate the combinations of adjacent positions:

3. $\sum_{o=1}^{i-1} A_{ok} + A_{ij} \geq 2 \leftrightarrow B_{ijk} = 1$

$$\sum_{o=1}^{i-1} A_{ok} + A_{ij} \geq 2 \times B_{ijk} \quad \forall i, U1(j,k)$$

$$\sum_{o=1}^{i-1} A_{ok} + A_{ij} \leq 1 + B_{ijk} \quad \forall i, U1(j,k)$$

4. $\sum_{o=1}^{i-1} A_{ok} + \sum_{o=1}^{i-1} A_{ol} + A_{ij} \geq 3 \leftrightarrow C_{ijkl} = 1$

$$\sum_{o=1}^{i-1} A_{ok} + \sum_{o=1}^{i-1} A_{ol} + A_{ij} \geq 3 \times C_{ijkl} \quad \forall i, U2(j,k,l)$$

$$\sum_{o=1}^{i-1} A_{ok} + \sum_{o=1}^{i-1} A_{ol} + A_{ij} \leq 2 + C_{ijkl} \quad \forall i, U2(j,k,l)$$

5. $\sum_{o=1}^{i-1} A_{ok} + \sum_{o=1}^{i-1} A_{ol} + \sum_{o=1}^{i-1} A_{om} + A_{ij} \geq 4 \leftrightarrow D_{ijklm} = 1$

$$\sum_{o=1}^{i-1} A_{ok} + \sum_{o=1}^{i-1} A_{ol} + \sum_{o=1}^{i-1} A_{om} + A_{ij} \geq 4 \times D_{ijklm} \quad \forall i, U3(j,k,l,m)$$

$$\sum_{o=1}^{i-1} A_{ok} + \sum_{o=1}^{i-1} A_{ol} + \sum_{o=1}^{i-1} A_{om} + A_{ij} \leq 3 + D_{ijklm} \quad \forall i, U3(j,k,l,m)$$

6. $\sum_{o=1}^{i-1} A_{ok} + \sum_{o=1}^{i-1} A_{ol} + \sum_{o=1}^{i-1} A_{om} + \sum_{o=1}^{i-1} A_{on} + A_{ij} \geq 5 \leftrightarrow E_{ijklmn} = 1$

$$\sum_{o=1}^{i-1} A_{ok} + \sum_{o=1}^{i-1} A_{ol} + \sum_{o=1}^{i-1} A_{om} + \sum_{o=1}^{i-1} A_{on} + A_{ij} \geq 5 \times E_{ijklmn} \quad \forall i, U4(j,k,l,m,n)$$

$$\sum_{o=1}^{i-1} A_{ok} + \sum_{o=1}^{i-1} A_{ol} + \sum_{o=1}^{i-1} A_{om} + \sum_{o=1}^{i-1} A_{on} + A_{ij} \leq 4 + E_{ijklmn} \quad \forall i, U4(j,k,l,m,n)$$

7. Calculation of points per turn and position:

$$p_{i1} = A_{i1} + B_{i12} + B_{i14} + C_{i123} + C_{i124} + C_{i147} \quad \forall i$$

$$p_{i2} = A_{i2} + B_{i21} + B_{i23} + B_{i25} + C_{i258} + C_{i215} + C_{i235} - D_{i2135} \quad \forall i$$

$$p_{i3} = A_{i3} + B_{i32} + B_{i36} + C_{i312} + C_{i369} + C_{i326} \quad \forall i$$

$$p_{i4} = A_{i4} + B_{i41} + B_{i45} + B_{i47} + C_{i456} + C_{i415} + C_{i457} - D_{i4157} \quad \forall i$$

$$p_{i5} = A_{i5} + B_{i52} + B_{i54} + B_{i56} + B_{i58} + C_{i524} + C_{i526} + C_{i548} + C_{i568} - D_{i5246} - D_{i5268} - D_{i5248} - D_{i5468} + E_{i52468} \quad \forall i$$

$$p_{i6} = A_{i6} + B_{i63} + B_{i65} + B_{i69} + C_{i645} + C_{i635} + C_{i659} - D_{i6359} \quad \forall i$$

$$p_{i7} = A_{i7} + B_{i74} + B_{i78} + C_{i714} + C_{i748} + C_{i789} \quad \forall i$$

$$p_{i8} = A_{i8} + B_{i85} + B_{i87} + B_{i89} + C_{i825} + C_{i857} + C_{i859} - D_{i8579} \quad \forall i$$

$$p_{i9} = A_{i9} + B_{i96} + B_{i98} + C_{i936} + C_{i968} + C_{i978} \quad \forall i$$

8. Total number of points:

$$TP = \sum_{i,j} p_{ij}$$

E. *Objective Function*

Maximization of the total number of points:

$$\max\{TP\}$$

IV. CODE

```
$TITLE AZUL

*Azul is a board game in which the goal is to obtain the maximum
number of
*points as a square is sequentially filled in. With this program we
attempt
*to maximize the total number of points.

*Jorge Bennasar Vázquez and Enrique Alonso Álvarez

*21/10/2018

OPTION optcr = 0

SETS
i                   turn / 1 * 9 /
j                   position / 1 * 9 /

ALIAS(i,o)
ALIAS(j,k,l,m,n)

SETS
U1(j,k)             connections between j and k
/1.2, 1.4, 2.1, 2.3, 2.5, 3.2, 3.6, 4.1, 4.5, 4.7, 5.2, 5.4, 5.6,
5.8, 6.3, 6.5, 6.9, 7.4, 7.8, 8.5, 8.7, 8.9, 9.6, 9.8/
U2(j,k,l)           connections between j k and l
/1.2.3, 1.2.4, 1.4.7, 2.5.8, 2.1.5, 2.3.5, 3.1.2, 3.6.9, 3.2.6,
4.5.6, 4.1.5, 4.5.7, 5.2.4, 5.2.6, 5.4.8, 5.6.8, 6.4.5, 6.3.5, 6.5.9,
7.1.4, 7.4.8, 7.8.9, 8.2.5, 8.5.7, 8.5.9, 9.3.6, 9.6.8, 9.7.8/
U3(j,k,l,m)         connections between j k l and m
/2.1.3.5, 4.1.5.7, 5.2.4.6, 5.2.6.8, 5.2.4.8, 5.4.6.8, 6.3.5.9,
8.5.7.9/
U4(j,k,l,m,n)       connections between j k l m and n
/5.2.4.6.8/

VARIABLES
TP                  total points
p(i,j)              points obtained for completing position j in turn i
A(i,j)              position j completed in turn i
B(i,j,k)            position k completed before j is completed in turn i
C(i,j,k,l)          positions k and l completed before j is completed in
turn i
D(i,j,k,l,m)        positions k l and m completed before j is completed
in turn i
E(i,j,k,l,m,n)      positions k l m and n completed before j is
completed in turn i
```

```
POSITIVE VARIABLES p;
BINARY VARIABLES A, B, C, D, E;

EQUATIONS
OF                      objective function: total points
OTPP(j)                 one turn per position
OPPT(i)                 one position per turn
AUX11(i,j,k)            auxiliary equation 11
AUX12(i,j,k)            auxiliary equation 12
AUX21(i,j,k,l)          auxiliary equation 21
AUX22(i,j,k,l)          auxiliary equation 22
AUX31(i,j,k,l,m)        auxiliary equation 31
AUX32(i,j,k,l,m)        auxiliary equation 32
AUX41(i,j,k,l,m,n)      auxiliary equation 41
AUX42(i,j,k,l,m,n)      auxiliary equation 42
P1(i)                   points 1
P2(i)                   points 2
P3(i)                   points 3
P4(i)                   points 4
P5(i)                   points 5
P6(i)                   points 6
P7(i)                   points 7
P8(i)                   points 8
P9(i)                   points 9;

OF..                    TP =E= SUM[(i,j),p(i,j)];

OTPP(j)..               1 =E= SUM[i,A(i,j)];
OPPT(i)..               1 =E= SUM[j,A(i,j)];

AUX11(i,j,k)$U1(j,k)..
                        2*B(i,j,k) =L= SUM[o $ (ORD(o) lt
ORD(i)),A(o,k)] + A(i,j);

AUX12(i,j,k)$U1(j,k)..
                        1+B(i,j,k) =G= SUM[o $ (ORD(o) lt
ORD(i)),A(o,k)] + A(i,j);

AUX21(i,j,k,l)$U2(j,k,l)..
                        3*C(i,j,k,l) =L= SUM[o $ (ORD(o) lt
ORD(i)),A(o,k)] + SUM[o $ (ORD(o) lt ORD(i)),A(o,l)] + A(i,j);

AUX22(i,j,k,l)$U2(j,k,l)..
                        2+C(i,j,k,l) =G= SUM[o $ (ORD(o) lt
ORD(i)),A(o,k)] + SUM[o $ (ORD(o) lt ORD(i)),A(o,l)] + A(i,j);

AUX31(i,j,k,l,m)$U3(j,k,l,m)..
                        4*D(i,j,k,l,m) =L= SUM[o $ (ORD(o) lt
ORD(i)),A(o,k)] + SUM[o $ (ORD(o) lt ORD(i)),A(o,l)] + SUM[o $
(ORD(o) lt ORD(i)),A(o,m)] + A(i,j);

AUX32(i,j,k,l,m)$U3(j,k,l,m)..
                        3+D(i,j,k,l,m) =G= SUM[o $ (ORD(o) lt
ORD(i)),A(o,k)] + SUM[o $ (ORD(o) lt ORD(i)),A(o,l)] + SUM[o $
(ORD(o) lt ORD(i)),A(o,m)] + A(i,j);

AUX41(i,j,k,l,m,n)$U4(j,k,l,m,n)..
                        5*E(i,j,k,l,m,n) =L= SUM[o $ (ORD(o) lt
ORD(i)),A(o,k)] + SUM[o $ (ORD(o) lt ORD(i)),A(o,l)] + SUM[o $
(ORD(o) lt ORD(i)),A(o,m)] + SUM[o $ (ORD(o) lt ORD(i)),A(o,n)] +
A(i,j);

AUX42(i,j,k,l,m,n)$U4(j,k,l,m,n)..
                        4+E(i,j,k,l,m,n) =G= SUM[o $ (ORD(o) lt
ORD(i)),A(o,k)] + SUM[o $ (ORD(o) lt ORD(i)),A(o,l)] + SUM[o $
(ORD(o) lt ORD(i)),A(o,m)] + SUM[o $ (ORD(o) lt ORD(i)),A(o,n)] +
A(i,j);

P1(i)..                 p(i,'1') =E= A(i,'1') + B(i,'1','2') +
B(i,'1','4') + C(i,'1','2','3') + C(i,'1','2','4') +
C(i,'1','4','7');

P2(i)..                 p(i,'2') =E= A(i,'2') + B(i,'2','1') +
B(i,'2','3') + B(i,'2','5') + C(i,'2','5','8') + C(i,'2','1','5') +
C(i,'2','3','5') - D(i,'2','1','3','5');

P3(i)..                 p(i,'3') =E= A(i,'3') + B(i,'3','2') +
B(i,'3','6') + C(i,'3','1','2') + C(i,'3','6','9') +
C(i,'3','2','6');

P4(i)..                 p(i,'4') =E= A(i,'4') + B(i,'4','1') +
B(i,'4','5') + B(i,'4','7') + C(i,'4','5','6') + C(i,'4','1','5') +
C(i,'4','5','7') - D(i,'4','1','5','7');

P5(i)..                 p(i,'5') =E= A(i,'5') + B(i,'5','2') +
B(i,'5','4') + B(i,'5','6') + B(i,'5','8') + C(i,'5','2','4') +
C(i,'5','2','6') + C(i,'5','4','8') + C(i,'5','6','8') -
D(i,'5','2','4','6') - D(i,'5','2','6','8') - D(i,'5','2','4','8') -
D(i,'5','4','6','8') + E(i,'5','2','4','6','8');

P6(i)..                 p(i,'6') =E= A(i,'6') + B(i,'6','3') +
B(i,'6','5') + B(i,'6','9') + C(i,'6','4','5') + C(i,'6','3','5') +
C(i,'6','5','9') - D(i,'6','3','5','9');

P7(i)..                 p(i,'7') =E= A(i,'7') + B(i,'7','4') +
B(i,'7','8') + C(i,'7','1','4') + C(i,'7','4','8') +
C(i,'7','8','9');

P8(i)..                 p(i,'8') =E= A(i,'8') + B(i,'8','5') +
B(i,'8','7') + B(i,'8','9') + C(i,'8','2','5') + C(i,'8','5','7') +
C(i,'8','5','9') - D(i,'8','5','7','9');

P9(i)..                 p(i,'9') =E= A(i,'9') + B(i,'9','6') +
B(i,'9','8') + C(i,'9','3','6') + C(i,'9','6','8') +
C(i,'9','7','8');

MODEL AZUL /all/;

SOLVE AZUL MAXIMIZING TP USING MIP
```

V. RESULTS

Before showing the results it is relevant to say that we had to use NEOS Server to solve the problem due to the complexity of it.

The maximum possible number of total point obtained with the model proposed is $TP = 33$. The matrix of positions completed per turn (A_{ij}) is shown below (Table 1):

	P1	P2	P3	P4	P5	P6	P7	P8	P9
T1	0	0	0	0	1	0	0	0	0
T2	0	0	0	0	0	0	1	0	0
T3	0	0	1	0	0	0	0	0	0
T4	0	0	0	1	0	0	0	0	0
T5	0	1	0	0	0	0	0	0	0
T6	1	0	0	0	0	0	0	0	0
T7	0	0	0	0	0	1	0	0	0
T8	0	0	0	0	0	0	0	1	0
T9	0	0	0	0	0	0	0	0	1

Table 1. Positions completed per turn.

Additionally, Table 2 shows the number of points obtained per turn and position (p_{ij}):

	P1	P2	P3	P4	P5	P6	P7	P8	P9
T1	0	0	0	0	1	0	0	0	0
T2	0	0	0	0	0	0	1	0	0
T3	0	0	1	0	0	0	0	0	0
T4	0	0	0	4	0	0	0	0	0
T5	0	4	0	0	0	0	0	0	0
T6	6	0	0	0	0	0	0	0	0
T7	0	0	0	0	0	5	0	0	0
T8	0	0	0	0	0	0	0	5	0
T9	0	0	0	0	0	0	0	0	6

Table 2. Points obtained per position and turn.

As we can see, the individual points summed up (1+1+1+4+4+6+5+5+6 = 33) equals TP.

The results of variables $B_{ijk}, C_{ijkl}, D_{ijklm}$ and E_{ijklmn} are not shown in the report due to the complexity of them. This variables have helped the program to detect the columns and rows created with each turn, and therefore have made possible the proper operation of the code (without them the points would not have been assigned correctly).

The total number of iterations to obtain the optimal solution was of 90694 with a relative tolerance of 0.

VI. EXTENSION OF THE CASE STUDY

In this section we have tried a specific modification of the problem: the program now has to complete position 5 in turn 9. The changes in the program in GAMS are not significant enough to show them in the report. We expect that the solution matrix p_{ij} will be symmetrical. Obviously, the optimal solution

TP now has to be inferior to the previous one (33).

In this case, the optimal total punctuation obtained was $TP = 30$. The points obtained per position and turn (p_{ij}) are shown below (Table 3):

	P1	P2	P3	P4	P5	P6	P7	P8	P9
T1	0	0	0	0	0	1	0	0	0
T2	0	0	0	1	0	0	0	0	0
T3	0	0	0	0	0	0	0	1	0
T4	0	1	0	0	0	0	0	0	0
T5	0	0	0	0	0	0	0	0	4
T6	4	0	0	0	0	0	0	0	0
T7	0	0	0	0	0	0	6	0	0
T8	0	0	6	0	0	0	0	0	0
T9	0	0	0	0	6	0	0	0	0

Table 3. Points obtained per position and turn.

As we expected, in this case we have symmetry in the matrix around position 5 (6-4, 8-2, 9-1, 7-3 and 5).

VII. CONCLUSIONS

In conclusion, the results obtained with the optimization model (Mixed Integer Linear) have been satisfactory and coherent with what we previously expected. The optimal punctuation in a 3x3 board of Azul is $TP = 33$. Additionally, changing the game so that we have to complete the central position (5) in the last turn (9) we obtain an optimal solution of $TP = 30$ and the points per position and turn matrix (p_{ij}) is now symmetrical.